UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

Basic program "IPFLTR" for induced polarization data reduction and filtering

by

Hamdy S. Sadek 1/

Open-File Report 83- 485

Prepared for the Ministry of Petroleum and Mineral Resources
Deputy Ministry for Mineral Resources
Jiddah, Kingdom of Saudi Arabia

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards and stratigraphic nomenclature.

1/ U.S. Geological Survey Mission, Jiddah, Saudi Arabia

1983

CONTENTS

Page

ABSTRACT.. 1

INTRODUCTION.................................... 1

ALGORITHM....................................... 2

Reduction.................................... 2
Filtering.................................... 6

COMPUTER PROGRAM................................ 6

Description.................................. 6
Operation.................................... 9

ACKNOWLEDGMENTS AND DATA STORAGE................. 19

REFERENCES CITED................................ 20

APPENDIX 1. Program listing..................... 21
2. Examples of program operation............ 26

ILLUSTRATIONS

Figure 1. Method of construction of dipole-dipole pseudosections........................ 3

2. Types of filters that can be used to produce contour maps of IP data........................ 7

3. Pseudosections showing sequence of input data to the computer program.............. 10

4. Flow charts showing Key mode 1 (input key)............ 11
5. Flow charts showing Key mode 2 (data storage key)..... 13
6. Flow charts showing Key mode 3 (data printout key).... 14
7. Flow charts showing Key mode 4 (input data correction key).......................... 15
8. Flow charts showing Key mode 5 (data reduction key)... 16
9. Flow charts showing Key mode 6 (filtering key)........ 17
10. Flow charts showing Key mode 7 (x-y plot key)......... 18

BASIC PROGRAM "IPFLTR" FOR INDUCED POLARIZATION DATA REDUCTION AND FILTERING

by

Hamdy S. Sadek 1/

ABSTRACT

The recently developed induced polarization (IP) mapping technique can be successfully applied to IP dipole-dipole data and includes three types of filters that can be applied to IP pseudosections that have IP anomalies caused by different types of sources. The technique yields a single output value per station; this value reflects all levels of the pseudosection and is suitable for contouring in plan view. It also provides a quantitative picture of IP anomalies in their background or regional settings.

In the "IPFLTR" program, the application of the three types of IP filters to the IP pseudosection data was programed in BASIC. The program allows the user to reduce IP field data and to produce IP pseudosections. It includes three filter subprograms, which calculate output data from each filter. The filtered data include a single reading for each station along the pseudosection. These data can be presented as individual IP profiles, or several profiles can be contoured together to produce an IP map. The program can optionally provide simple or complex plots either of the filtered data or of any selected level in the pseudosection.

INTRODUCTION

The dipole-dipole induced polarization (IP) technique has been described in most geophysical exploration text books (see, for example, Telford and others, 1975, p. 702-735). IP data are generally presented as either pseudosections or multilevel profiles. In most cases, these methods of presentation are not desirable because the most anomalous values are displaced laterally and do not coincide with the source body.

A new technique of IP mapping (Fraser, 1981) provides a qualitative presentation of dipole-dipole IP data in map form and includes three averaging filters suitable for different types of IP sources. Any selected filter yields a single value per station; this value reflects data from all levels of the pseudosection. If these values are plotted, the most anomalous values calculated from the pseudosections overlie the source body.

Because IP data produced in a survey along a single profile are generally presented in at least three pseudosections

1/ U.S. Geological Survey Mission, Jiddah, Saudi Arabia

(apparent resistivity, percent frequency effect or chargeability, and metal factor), the manual application of these filters to the pseudosections of a surveyed IP profile is time consuming and impractical.

For this reason, a computer program was designed to carry out the following operations:

1. Reduce IP field results and calculate pseudosections of either resistivity, percent frequency effect or chargeability, metal factor, or phase.

2. Apply any of the three filters suggested by Fraser (1981).

3. Allow single or overlapped x-y plotting of the filtered data or any selected level in a particular pseudosection.

ALGORITHM

Reduction

The computer program allows the user to reduce dipole-dipole IP field data to yield the quantities: resistivity (ρ), percent frequency effect (PFE) or chargeability (M), and metal factor (MF). The potential difference, ΔV, measured between two-receiver electrodes (fig. 1) when a current I is passed between the two current electrodes is given by

$$\Delta V = \frac{I\rho}{2\pi G}$$

where

ρ is the resistivity of the ground and
G is the geometric factor

$$G = \frac{1}{\left(\frac{1}{n} - \frac{2}{n+1} - \frac{1}{n+2}\right)}$$

where n represents the number of IP levels.

Setting

$$K = 2\pi Ga$$

where a is the dipole length, then

$$\rho = K\left(\frac{\Delta V}{I}\right),$$

which is Ohm's law. In this case, K is known as the K factor



Figure 1.--Diagram showing method of construction of dipole-dipole pseudosections for N=2, plot point P and first-station location S_1. N, level number; M, dipole number; a, dipole spacing.

(Telford and others, 1975, p. 716). If V is expressed in millivolts, then K is divided by 1,000.

The computer program enables the user to reduce the following three types of IP dipole-dipole field data.

1. <u>Time-domain IP</u>

If a direct-current pulse, I, is transmitted between the two current electrodes (fig. 1), then the measured quantities at the receiver will be ΔV (primary received voltage) and M. The direct-current resistivity, ρ_{DC}, is given by

$$\rho_{DC} = K(\frac{\Delta V}{I})$$

and the metal factor is expressed as

$$MF = 1000 \times M/\rho_{DC}$$

2. <u>Frequency-domain IP</u>

(a) Standard frequency-domain IP

The first-developed and most common method of IP measurement in the frequency domain entails the comparison of earth resistivities at two widely spaced frequencies, often a decade apart. Such comparison requires measurements of the currents, I_{f1} and I_{f2}, and voltages, V_{f1} and V_{f2}, by use of sequential transmission of current at frequencies f1 and f2, respectively. The apparent resistivities at frequencies f1 and f2 are given as

$$\rho_{af1} = K(\frac{\Delta V_{f1}}{I_{f1}}), \text{ and}$$

$$\rho_{af2} = K(\frac{\Delta V_{f2}}{I_{f2}}).$$

The traditional frequency-domain method produces percent-frequency effect as the measured quantity (Telford and others, 1975). This quantity is equivalent to chargeability in the time-domain IP and is defined as

$$PFE = \frac{\rho_{af1} - \rho_{af2}}{\rho_{af1}}.$$

According to Parasnis (1979), the metal factor is given either as

$$MF = 1000\ PFE/\rho_a$$

or as

$$MF = 1000\ PFE/(\rho_a/2\pi).$$

(b) Phase IP

More recently, a second method of IP measurement in the frequency domain has come into use; in this method, the phase angle, ϕ, between the measured voltage in the ground and the primary current is measured. Zong and others (1972) and Hallof (1974) showed the general equivalence between the phase angle measured at a specific frequency in the IP range and the percent-frequency effect measured at two frequencies spanning this range. Moreover, they were able to use a single frequency to obtain percent frequency effect rather than two sequential frequencies, as in standard IP frequency-domain techniques.

In the phase-IP method, the quantities actually measured by using specifically manufactured receivers, such as the Scintrex IPRF2*, are as follows.

(i) The apparent resistivity, ρ_a, using single frequency, defined as

$$\frac{K(\Delta V)}{I}$$

(ii) The PFE, defined as

$$\left(\frac{A_1-3A_3}{3A_3}\right)100$$

where A_1 and A_3 are the amplitudes of the fundamental and third harmonic ground signal, respectively.

(iii) The phase defined as the phase angle between the measured voltage, ΔV, and the primary current, I.

Some receivers also measure the relative phase shift (RPS). This quantity is defined as

$$3\phi_1 - \phi_3$$

where ϕ_1 and ϕ_3 are the phase angle of the fundamental and third harmonic, respectively, relative to the transmitted component of the current.

(iv) The metal factor calculated as

$$MF = 1000 \times PFE/(\rho_a/2\pi)$$

*Use of specific manufacturers' names does not constitute endorsement by the U.S. Geological Survey and is used for information purposes only.

Filtering

Fraser (1981) suggested three types of IP filters, each of which provides a single IP value per station for any given dipole-dipole IP pseudosection.

1. Average levels filter

This filter (fig. 2A) is a simple averaging filter that is applied to each level. The average of two adjacent values is taken for the second level, that of three values for the third level, and so forth. The first level remains the same. The shape of the window filter was chosen to crudely approximate the pseudosection anomaly pattern that commonly results from thin, steeply dipping bodies. Final output for each point is the average of the averages for all levels.

2. Total average filter

This filter is equivalent to a simple application of the arithmetic average of all values in the triangular window (fig. 2B). Because more points come from the deeper levels, the output of this filter is progressively weighted toward the deeper levels of the pseudosection. This shift could be an advantage if the target body is evident only on the deeper levels.

3. Average of triangle sides filter

This filter provides a simple arithmetic average of all the values in the window (fig. 2C). Its primary disadvantage is that its output can oscillate if a single large value occurs on one of the deeper levels of the input pseudosection.

The program allows the user to move any one of these three filters over any given IP pseudosection and to obtain a single IP value per station. If some stations within a window have no values, for example, at the ends of lines where data are typically absent on the deeper levels (fig. 2), then the average of fewer values will be considered.

COMPUTER PROGRAM

Description

The program (appendix 1) was written specifically for use with the Hewlett Packard (HP) model 85. This machine is extensively used either as an individual minicomputer or coupled with other devices such as field geophysical equipment. The program was written in BASIC and can be modified to operate on any other machine that uses the same computer language. Conversion of the plotting subprograms may be some-



Figure 2.--Diagram showing the types of filters that can be used to produce contour maps and IP data: (A) average levels filter; (B) total average filter; (C) average of triangle sides filter. Diagram modified from Fraser, 1981.

what difficult, inasmuch as many of the plotting commands are not unique in all devices.

The program is conversational and allows the user a wide degree of latitude as to data input, filters, and plotting, simply by answering prompted questions.

The pseudosection, the filtered data, and a simple or complex x-y plot of the previous data are computed during program execution.

The input/output (I/O) quantities are as follows:

Numeric constants

M: number of dipoles along surveyed profile (index J = 1,....,M) (M max = 50)

N: number of levels in a pseudosection (index I = 1,....,N) (N max = 6)

S1: first station value, numeric value, south and west values are negative

A: dipole spacing, in meters or feet

W1: =1 for (average levels filter); =2 for (total average filter); =3 for 2 triangle sides filter

W2: =1 for resistivity; =2 for percent frequency effect or chargeability; =3 for metal factor; =4 for phase

Strings

A$: string of area name

L$: string of line name

T$: string of type of survey; frequency (F) or time (T) domain

O$: station order, increase or decrease from the value of S1

Arrays

R (I,J): array of resistivity pseudosection data

F (I,J): array of percent frequency effect or chargeability pseudosection data

K (I,J): array of metal-factor pseudosection data

P (I,J): array of phase or phase-shift pseudosection data

W (J): array of filtered data

Z (I,J): working array

where J=1,...,M and I=1,...,N

The maximum number of dipoles (M) is 50, and the maximum number of levels (N) is 6. These values are suitable for use with most of IP field surveys. However, these values can be changed if the computer used has sufficient memory capability.

Operation

The program is operated using eight independent key modes. Each key mode includes one or more specially designed subprograms. The user can carry out one specific operation per key mode, such as data input, filtering, printing, storage, and so forth.

1. Key mode 1 (input key)

Key mode 1 allows the user to input IP data from either the keyboard or a tape file. The subprogram prompts the user for file name, area, line, and other information used to identify the input data. The input IP data itself may exist in any of the following forms:

(a) Reduced IP pseudosection data (time or frequency domain). These will be stored in R, F, K and P arrays, for resistivity, chargeability or percent frequency effect, metal factor, and phase or relative phase shift, respectively.

(b) Nonreduced IP pseudosection data (time or frequency domain) obtained from field surveys. These will be stored in R, F, K or P arrays.

(c) Filtered profile data obtained from any of the R, F, K or P pseudosections. These will be stored in array W.

An explanation of IP data input entry is shown on figure 3. The operation of key mode 1 is illustrated on the flow chart (fig. 4), and an example is provided in appendix 2.



Figure 3.--Pseudosections showing sequence of input data to the computer program: R, apparent resistivity; F, chargeability or percent frequency effect, K, metal factor; and P, phase shift or phase. See text for an explanation of the quantities J, M, I, and N.



Figure 4.--Flow chart showing key mode 1 (input key).

2. <u>Key mode 2 (data storage key)</u>

Key mode 2 allows data storage on a specific file of a magnetic tape. By use of this key mode, nonreduced field IP data, pseudosection data, or filtered data can be stored.

The operation of key mode 2 is illustrated on the flow chart (fig. 5), and an example is provided in appendix 2.

3. <u>Key mode 3 (data print-out key)</u>

Key mode 3 allows the user to a print out any of the three previously mentioned data forms. Key 3 mode is illustrated in the flow chart (fig. 6), and an example is shown in appendix 2.

4. <u>Key mode 4 (input data correction key)</u>

Key mode 4 allows the user to correct the various types of the input data.

Key mode 4 is explained in the flow chart (fig. 7), and an example is given in appendix 2.

5. <u>Key mode 5 (data reduction key)</u>

Key mode 5 allows the user to reduce IP field data, which may exist in one of the following forms:

(a) Time-domain IP data, which includes values of Vp (voltage) and I (current) for calculation of direct-current resistivity and metal factor.

(b) Frequency-domain IP data, which are measured by using one frequency and include phase-shift measurements. The apparent resistivity and metal factor are also calculated by using a single set of Vp and I values.

(c) Standard frequency-domain IP data, which are measured by using two different frequencies; at each frequency, Vp and I values are measured. Two apparent resistivity values and the metal factor are calculated.

The key mode is explained in the flow chart (fig. 8), and an example is shown in appendix 2.

6. <u>Key mode 6 (filtering key)</u>

Key mode 6 allows the user to obtain filtered data for any selected IP pseudosection. The mode includes three sub-



Figure 5.--Flow chart showing key mode 2 (data storage key).



Figure 6.--Flow chart showing key mode 3 (data print-out key).



Figure 7.--Flow chart showing key mode 4 (input data correction key).



Figure 8.--Flow chart showing key mode 5 (data reduction key).



Figure 9.--Flow chart showing key mode 6 (filtering key).



Figure 10.--Flow chart showing key mode 7 (x-y plot key).

programs, each of which allow the application of one of the three filters suggested by Fraser (1981). The output-filtered data are printed automatically during operation of the key. Key mode 6 is explained in the flow chart (fig. 9), and an example is shown in appendix 2.

7. Key mode 7 (x-y plot key)

Key mode 7 allows the user to produce either a simple or complex x-y plot of either any level of any specific pseudosection or any filtered data stored in array W.

The plot is produced on the screen, and a copy can be made on the thermal printer. The key mode allows the user a wide range of plot specifications in both the x- and y-dimensions.

The operation of key mode 7 is illustrated on the flow chart (fig. 10), and an example is shown in appendix 2.

8. Key mode 8 (stop key)

Key mode 8 allows the user to stop program execution.

ACKNOWLEDGMENTS AND DATA STORAGE

The work on which this report is based was performed in accordance with a work agreement between the U.S. Geological Survey (USGS) and the Saudi Arabian Ministry of Petroleum and Mineral Resources. No base data files were established for this report, and no entries were made to the Mineral Occurrence Documentation System (MODS) as a result of this report.

REFERENCES CITED

Fraser, D. C., 1981, Contour map presentation of dipole-dipole induced polarization data: Geophysical Prospecting, v. 29, p. 639-651.

Hallof, P., 1974, The IP phase measurement and inductive coupling: Geophysics, v. 39, p. 650-665.

Parasnis, D. S., 1979, Principles of applied geophysics: London, Champan and Hall, 275 p.

Telford, W. M., Geldart, L. P., Sheriff, R. E., and Keys, D. A., 1975, Applied geophysics: New York-London, Cambridge University Press, 877 p.

Zong, K. L., Samck, W. A. and Sumner, J. S., 1972, Comparison of time, frequency and phase measurements in induced polarization: Geophysical Prospecting, v. 20, p. 626-648.

Appendix 1.--Program listing

```
10 ! PROGRAM IPFLTR
20 OPTION BASE 1
30 DIM R(6,50),F(6,50),M(6,50),P
(6,50),C(6),W(50),Z(6,50),F2$[6]
,F3$[6]
40 PRINT "IP DATA CALCULATIONS"
50 PRINT "==================="
60 CLEAR
70 DISP "  "
80 DISP "  "
90 DISP "SELECT THE MODE OF OPER
ATION BY"
100 DISP "
          THE KEY BELOW (ONE AT
ATIME)"
110 ON KEY# 1,"INPUT" GOSUB 210
120 ON KEY# 2,"STORE" GOSUB 2080

130 ON KEY# 3,"PRINT" GOSUB 1760

140 ON KEY# 4,"MISTAKES" GOSUB 1
230
150 ON KEY# 5,"REDUCE" GOSUB 150
0
160 ON KEY# 6,"FILTER" GOSUB 235
0
170 ON KEY# 7,"PLOT" GOSUB 3940
180 ON KEY# 8,"STOP" GOSUB 4740
190 KEY LABEL
200 GOTO 110
210 ! SUBPROGRAM INPUT
220 CLEAR
230 DISP "THIS SUBPROGRAM ALLOWS
 DATA INPUT FROM TAPE OR KEYBOAR
D"
240 DISP "OPERATE KEY?(Y/N):"
250 INPUT K1$
260 IF K1$="N" THEN 1200
270 DISP "INPUT FROM TAPE (T) OR
 KEYBOARD(K):"
280 INPUT I1$
290 IF I1$="T" THEN 1020
300 DISP "AREA NAME?:"
310 INPUT A$
320 DISP "LINE NAME?:"
330 INPUT L$
340 DISP "TYPE OF SURVEY, TIME D
OMAIN(T) OR FREQ. DOMAIN(F)?"
350 INPUT T$
360 IF T$="F" THEN T1$="PFE" ELS
E T1$="CHRG"
370 DISP "STATION LOCATION WILL
BE -VE FOR S&W AND +VE FOR N&E"
380 DISP "FIRST STATION LOCAION
ON N=1 DIPOLE?:"
390 INPUT S1
400 DISP "STATION ORDER DECREASE
 (D) OR INCREASE (I)?:"
410 INPUT O$
420 DISP "NO. OF DIPOLES ALONG P
ROFILE?"
430 INPUT N
440 DISP "DIPOLE LENGTH?:"
450 INPUT D
460 DISP "NO OF 1P LEVELS(N=)?:"

470 INPUT M
480 D1=D
490 IF O$="D" THEN 520
500 S2=S1+3*D/2
510 GOTO 530
520 S2=S1-3*D/2 @ D=-D
530 IF T$="T" THEN PRINT " TIME
DOMAIN IP SURVEY"
540 IF T$="F" THEN PRINT " FREQU
ENCY DOMAIN IP SURVEY"
550 PRINT "AREA:";A$
560 PRINT "LINE:";L$;" FIRST ST
ATION:";S2
570 PRINT "NO. OF DIPOLES=";N;"
 N=";M
580 PRINT "DIPOLE LENGTH:";D
590 PRINT "----------------"
600 IF I1$="T" THEN 1160
610 DISP "INPUT DATA REDUCED?(Y/
N):"
620 INPUT G$
630 IF T$="T" THEN D$="0" @ GOTO
 660
640 DISP "ONE FREQUENCY USED?(Y/
N)"
650 INPUT D$
660 PRINT "--INPUT DATA---"
670 DISP "INPUT PSEUDO(P) OR WIN
DOW(W)?" @ INPUT I2$
680 IF I2$="W" THEN 920
690 FOR J=1 TO N
700 DISP "DIPOLE NO.:";J
710 S3=S2+(J-1)*D
720 PRINT "DIPOLE NO=";J;" STATI
ON=";S3
730 IF G$="Y" THEN 770
740 IF T$="T" THEN PRINT "N=   V
P     CR      M" @ GOTO 780
750 IF D$="Y" THEN PRINT "N=
VP      CR     PFE   RPS" @ GOTO
780
760 IF D$="N" THEN PRINT "N=
VP1    CR1    VP2   CR2" @ GOTO
780
770 PRINT "N=    RES.  ";T1$;"
  MF    RPS"
780 FOR I=1 TO M
790 DISP "FOR LEVEL N=";I
800 IF G$="Y" THEN 840
810 IF T$="T" THEN DISP "INPUT V
P,CR,CHRG,RPS" @ GOTO 860
820 IF D$="Y" THEN DISP "INPUT V
p,Cr,PFE,RPS" @ GOTO 860
830 DISP "INPUT VP1,CR1,VP2,CR2"
 @ GOTO 860
840 IF T$="T" THEN DISP "INPUT R
,CHRG,MF,RPS" @ GOTO 860
850 DISP "INPUT R,PFE,MF,RPS"
860 INPUT R(I,J),F(I,J),M(I,J),P
(I,J)
```

```
870 PRINT USING 880 ; I,R(I,J),F
(I,J),M(I,J),P(I,J)
880 IMAGE DD,1X,DDDD.D,1X,DDDD.D
,1X,DDDD.D,1X,DDD.D
890 NEXT I
900 NEXT J
910 GOTO 1200
920 DISP "FILER OF AVER. LEVEL(1
) OR TOTAL AVERAGE(2)?" @ INPUT
W1
930 DISP "FILTERED DATA RES.(1),
PFE(2),MF(3) OR RPS(4)?" @ INPUT
 W2
940 PRINT "STATION             MA
P VALUE"
950 S2=S1+3*D/2
960 FOR J=1 TO N
970 S3=S2+D*(J-1)
980 DISP "INPUT READING NO.";J @
 INPUT W(J)
990 PRINT USING 2640 ; S3,W(J)
1000 NEXT J
1010 GOTO 1200
1020 DISP "INPUT FILENAME:"
1030 INPUT F2$
1040 ASSIGN# 2 TO F2$
1050 READ# 2,1 ; A$,L$,T$,D$,O$,
T1$,I2$,G$
1060 READ# 2,2 ; N,M,D,S1
1070 S2=S1+3*D/2
1080 IF I2$="W" THEN READ# 2,3 ;
 W1,W2,U1$
1090 FOR J=1 TO N
1100 IF I2$="W" THEN READ# 2 ; W
(J)@ GOTO 1140
1110 FOR I=1 TO M
1120 IF I2$="P" THEN READ# 2 ; R
(I,J),F(I,J),M(I,J),P(I,J)
1130 NEXT I
1140 NEXT J
1150 GOTO 490
1160 PRINT "INPUT DATA READ FROM
 FILE:";F2$
1170 IF I2$="W" THEN PRINT "INPU
T DATA FILTERED" @ GOTO 1200
1180 IF G$="Y" THEN PRINT "INPUT
 PSEUDO-SECTION"
1190 IF G$="N" THEN PRINT "INPUT
 FIELD DATA"
1200 CLEAR
1210 RETURN
1220 IMAGE DDD,1X,DDDD.D,1X,DDD.
D,1X,DDDD.D,2X,DDD.D
1230 ! SUBPROGRAM MISTAKES
1240 CLEAR
1250 DISP "THIS SUBPROGRAM ALLOW
S CORRECTION OF MISTAKES IN INPU
T DATA"
1260 DISP "OPERATE KEY?(Y/N):" @
 INPUT K1$
1270 IF K1$="N" THEN 1480
1280 DISP "ANY MISTAKES (Y/N)?"
1290 INPUT H$
1300 IF H$="N" THEN 1480
1310 DISP "MISTAKES IN PSEUDO?(Y
/N)" @ INPUT M2$
1320 IF M2$="N" THEN 1430
1330 DISP "WHICH DIPOLE(J) AND N
 (I)"
1340 INPUT J5,I5
1350 PRINT "CORRECTION OF INPUT
VALUE OF DIPOLE:";J5;"N=";I5
1360 PRINT "WRONG VALUE:"
1370 PRINT R(I5,J5);F(I5,J5);M(I
5,J5);P(I5,J5)
1380 DISP "ENTER NEW VALUES:"
1390 INPUT R(I5,J5),F(I5,J5),M(I
5,J5),P(I5,J5)
1400 PRINT "CORRECT VALUE:"
1410 PRINT R(I5,J5);M(I5,J5);F(I
5,J5);P(I5,J5)
1420 GOTO 1280
1430 DISP "WHICH ROW NO." @ INPU
T I5
1440 PRINT "WRONG VALUE:",W(I5)
1450 DISP "ENTER CORRECT VALUE"
@ INPUT W(I5)
1460 PRINT "CORRECT VALUE:",W(I5
)
1470 GOTO 1280
1480 CLEAR
1490 RETURN
1500 ! SUBPROGRAM REDUCION
1510 CLEAR
1520 DISP "THIS ROUTINE ALLOWS R
EDUCION OF PSEUDOSECTION DATA"
1530 DISP "OPERATE KEY?(Y/N)" @
INPUT K1$
1540 IF K1$="N" THEN 1740
1550 FOR I=1 TO M
1560 F7=1/I-2/(I+1)+1/(I+2) @ F7
=1/F7
1570 C(I)=2*PI*F7*D/1000
1580 NEXT I
1590 DISP "MULTIPLY METEAL FACTO
R BY 2PI?(Y/N)" @ INPUT M5$
1600 FOR J=1 TO N
1610 FOR I=1 TO M
1620 IF T$="T" THEN 1640
1630 IF D$="N" THEN 1660
1640 R(I,J)=R(I,J)*C(I)/F(I,J) @
 F(I,J)=M(I,J)
1650 M(I,J)=F(I,J)*1000/R(I,J) @
 GOTO 1710
1660 R1=C(I)*R(I,J)/F(I,J) @ R2=
C(I)*M(I,J)/P(I,J)
1670 R(I,J)=R1 @ P(I,J)=R2
1680 F(I,J)=100*(R1-R2)/R2
1690 M(I,J)=1000*F(I,J)/R1
1700 IF M5$="Y" THEN K(I,J)=K(I,
J)*2*PI
1710 NEXT I
1720 NEXT J
1730 G$="Y"
1740 CLEAR
1750 RETURN
1760 ! SUBPROGRAM PRINT
1770 CLEAR
1780 DISP "THIS ROUTINE PROVIDES
 PRINT OUTS OF PSEUDOSECTIONS OR
 FILTERED DATA "
1790 DISP "OPERATE KEY?(Y/N):" @
 INPUT K1$
```

```
1800 IF K1$="N" THEN 2060
1810 DISP "PRINT PSEUDO(P) OR WI
NDOW(W)"
1820 INPUT P2$
1830 IF P2$="W" THEN 1970
1840 IF G$="Y" THEN PRINT "PSEUD
O SECTION DATA"
1850 IF G$="N" THEN PRINT "IP FI
ELD DATA"
1860 FOR J=1 TO N
1870 S3=S2+(J-1)*D
1880 PRINT "DIPOLE:";J;" STATION
:";S3
1890 IF G$="Y" THEN PRINT "N=
  RES.   ";T1$;"     MF     RPS" @
GOTO 1920
1900 IF D$<>"N" THEN PRINT "N=
   VP      CR    ";T1$;"    RPS"
1910 IF D$="N" THEN PRINT "N=
 VP1     CR1    VP2      CR2"
1920 FOR I=1 TO M
1930 PRINT USING 1220 ; I,R(I,J)
,F(I,J),M(I,J),P(I,J)
1940 NEXT I
1950 NEXT J
1960 GOTO 2060
1970 PRINT "FILTERED DATA"
1980 IF W1=1 THEN PRINT "AVERAGE
 LEVELS FILTER OF LINE:";L$
1990 IF W1=2 THEN PRINT "TOTAL A
VERAGE FILTER OF LINE:";L$
2000 PRINT "-------------------"

2010 S2=S1+3*D/2
2020 FOR J=1 TO N
2030 S3=S2+(J-1)*D
2040 PRINT USING 2640 ; S3,W(J)
2050 NEXT J
2060 CLEAR
2070 RETURN
2080 ! SUBPROGRAM STORE DATA
2090 CLEAR
2100 DISP "THIS ROUTINE ALOWS ST
ORE ANY PART OR ALL OF INPUT AND
/OR OUTPUT DATA ON TAPE"
2110 DISP "OPERATE KEY?(Y/N):" @
 INPUT K1$
2120 IF K1$="N" THEN 2330
2130 DISP "OUTPUT FILENAME AND N
O. OF REC.?" @ INPUT F3$,R3
2140 CREATE F3$,R3 @ ASSIGN# 1 T
O F3$
2150 DISP "STORE PSEUDO(P) OR WI
NDOW(W) DATA?:" @ INPUT I2$
2160 PRINT# 1,1 ; A$,L$,T$,D$,O$
,T1$,I2$,G$
2170 PRINT# 1,2 ; N,M,D,S1
2180 IF I2$="W" THEN PRINT# 1,3
; W1,W2,U1$
2190 FOR J=1 TO N
2200 J1=J+3
2210 IF I2$="W" THEN PRINT# 1,J1
 ; W(J) @ GOTO 2250
2220 FOR I=1 TO M
2230 PRINT# 1 ; R(I,J),F(I,J),M(
I,J),P(I,J)
2240 NEXT I
2250 NEXT J
2260 PRINT "-------------------"
2270 PRINT "DATA STORED ON FILE:
";F3$
2280 IF I2$="W" THEN PRINT "FILT
ERED DATA OF" @ GOTO 2310
2290 IF G$="Y" THEN PRINT "REDUC
ED PSEUDOSECTION OF"
2300 IF G$="N" THEN PRINT "IP FI
ELD DATA OF"
2310 PRINT "AREA:";A$
2320 PRINT "LINE:";L$
2330 CLEAR
2340 RETURN
2350 ! SUBPROGRAM IP DATA FILTER

2360 CLEAR
2370 DISP "THIS ROUTINE ALLOWS A
PPLICATION OF THREE IP FILTERS"
@ DISP "SUGG BY FRASER (1981)"
2380 DISP "WINDOW (1) FOR AVER.
LEVELS (2) FOR TOTAL AVER AND (3
) FOR AVERAGE TRIANGLE SIDES"
2390 DISP "OPERATE KEY?(Y/N)" @
INPUT K1$
2400 IF K1$="N" THEN 2670
2410 DISP "APPLY MAP WINDOW?(Y/N
)"
2420 INPUT W$
2430 IF W$="N" THEN 2670
2440 DISP "TITLE OF DATA TO BE F
ILTERED"
2450 INPUT U1$
2460 DISP "AV. LEV.(1) OR ALL AV
.(2) OR TRIANGLE SIDES (3):"
2470 INPUT W1
2480 DISP "TO RES.(1),PFE(2),MF(
3) OR RPS(4)?"
2490 INPUT W2
2500 GOSUB 4780
2510 IF W1=1 THEN GOSUB 2690
2520 IF W1=2 THEN GOSUB 3140
2530 IF W1=3 THEN GOSUB 3530
2540 PRINT "FILTERED MAP RESULTS
 OF ";U1$
2550 IF W1=1 THEN PRINT "AVERAGE
 LEVELS FILTER"
2560 IF W1=2 THEN PRINT "TOTAL A
VERAGE FILTER"
2570 IF W1=3 THEN PRINT "AVERAGE
 SIDES FILTER"
2580 PRINT "-------------------"

2590 PRINT "STATION","MAP VALUE"

2600 S2=S1+3*D/2
2610 FOR I=1 TO N
2620 S3=S2+D*(I-1)
2630 PRINT USING 2640 ; S3,W(I)
2640 IMAGE DDDDD.D,15X,DDDDD.D
2650 NEXT I
2660 GOTO 2410
2670 CLEAR
2680 RETURN
2690 ! SUBPROGRAM IPWINDOW-1
2700 ! AVERAGE LEVELS FILTER
2710 FOR J=1 TO N
2720 B1=0
```

```
2730 IF J<M THEN 2870
2740 N2=N-M+1
2750 IF J>N2 THEN 2990
2760 FOR I=1 TO M
2770 I1=I
2780 B2=0 @ B3=0
2790 FOR K=1 TO I1
2800 J1=J-K+1
2810 B3=B3+Z(I,J1) @ B2=B2+1
2820 NEXT K
2830 V1=B3/B2 @ B1=B1+V1
2840 NEXT I
2850 W(J)=B1/M
2860 GOTO 3120
2870 FOR I=1 TO M
2880 I1=I
2890 IF I1>J THEN I1=J
2900 B2=0 @ B3=0
2910 FOR K=1 TO I1
2920 J1=J-K+1
2930 B3=B3+Z(I,J1) @ B2=B2+1
2940 NEXT K
2950 V1=B3/B2 @ B1=B1+V1
2960 NEXT I
2970 W(J)=B1/M
2980 GOTO 3120
2990 J1=N-J+1
3000 FOR I=1 TO M
3010 I1=I
3020 IF I1>J1 THEN I1=J1
3030 J2=J-I
3040 B2=0 @ B3=0
3050 FOR K=1 TO I1
3060 J3=J2+K
3070 B3=B3+Z(I,J3) @ B2=B2+1
3080 NEXT K
3090 V1=B3/B2 @ B1=B1+V1
3100 NEXT I
3110 W(J)=B1/M
3120 NEXT J
3130 RETURN
3140 ! SUBPROGRAM IPWINDOW-2
3150 ! TOTAL AVERAGE FILTER
3160 FOR J=1 TO N
3170 B2=0 @ B3=0
3180 IF J<M THEN 3300
3190 N2=N-M+1
3200 IF J>N2 THEN 3400
3210 FOR I=1 TO M
3220 I1=I
3230 FOR K=1 TO I1
3240 J1=J-K+1
3250 B3=B3+Z(I,J1) @ B2=B2+1
3260 NEXT K
3270 NEXT I
3280 W(J)=B3/B2
3290 GOTO 3510
3300 FOR I=1 TO M
3310 I1=I
3320 IF I1>J THEN I1=J
3330 FOR K=1 TO I1
3340 J1=J-K+1
3350 B3=B3+Z(I,J) @ B2=B2+1
3360 NEXT K
3370 NEXT I
3380 W(J)=B3/B2
3390 GOTO 3510
3400 FOR I=1 TO M
3410 I1=I
3420 J1=N-J+1
3430 IF I1>J1 THEN I1=J1
3440 J2=J-I
3450 FOR K=1 TO I1
3460 J3=J2+K
3470 B3=B3+Z(I,J3) @ B2=B2+1
3480 NEXT K
3490 NEXT I
3500 W(J)=B3/B2
3510 NEXT J
3520 RETURN
3530 ! SUBPROGRAM IPWINDOW3
3540 ! AVERAGE SIDES FILTER
3550 FOR J=1 TO N
3560 B2=0 @ B3=0
3570 IF J<M THEN 3700
3580 N2=N-M+1
3590 IF J>M THEN 3810
3600 FOR I=1 TO M
3610 IF I=1 THEN 3650
3620 J1=J-I+1
3630 B3=B3+Z(I,J1)
3640 B2=B2+1
3650 B3=B3+Z(I,J)
3660 B2=B2+1
3670 NEXT I
3680 W(J)=B3/B2
3690 GOTO 3920
3700 FOR I=1 TO M
3710 IF I=1 THEN 3760
3720 J1=J-I+1
3730 IF J1<1 THEN 3760
3740 B3=B3+Z(I,J1)
3750 B2=B2+1
3760 B3=B3+Z(I,J)
3770 B2=B2+1
3780 NEXT I
3790 W(J)=B3/B2
3800 GOTO 3920
3810 FOR I=1 TO M
3820 IF I=1 THEN 3870
3830 J1=N-I+1
3840 IF J>J1 THEN 3870
3850 B3=B3+Z(I,J)
3860 B2=B2+1
3870 J2=J-I+1
3880 B3=B3+Z(I,J2)
3890 B2=B2+1
3900 NEXT I
3910 W(J)=B3/B2
3920 NEXT J
3930 RETURN
3940 ! SUBPROGRAM PLOT
3950 CLEAR
3960 DISP "THIS ROUTINE ALOWS X-
Y PLOTS OF IP OUTPUT DATA ALONG
PROFILE IN LOG OR NORM. SCALE"
3970 DISP "OPERATE KEY?(Y/N)" @
INPUT K1$
3980 IF K1$="N" THEN 4650
3990 DISP "NORMAL(N) OR LOG(L) P
LOT?:" @ INPUT P6$
```

```
4000 DISP "PLOT WINDOW(W) OR PSE
UDO LEVEL(P)" @ INPUT P5$
4010 IF P5$="W" AND P6$="N" THEN
 4140
4020 IF P5$="W" AND P6$="L" THEN
 4060
4030 IF P5$="P" THEN DISP "LEVEL
 OF PSEUDO DATA" @ INPUT I5
4040 DISP "(P) TYPE RES(1),PFE(2
),MF(3) OR RPS(4)" @ INPUT W2
4050 GOSUB 4780
4060 FOR J=1 TO N
4070 IF P5$="P" THEN 4100
4080 IF W(J)=0 THEN 4130
4090 W(J)=LGT(W(J)) @ GOTO 4130
4100 W(J)=Z(I5,J)
4110 IF W(J)=0 THEN 4130
4120 IF P6$="L" THEN W(J)=LGT(W(
J))
4130 NEXT J
4140 GOSUB 4670
4150 M2=CEIL(M2)
4160 M1=FLOOR(M1)
4170 PRINT "--------------------"

4180 PRINT "PLOT OF LINE:";L$ @
PRINT "AREA:";A$
4190 PRINT "PLOT PARAMETERS:"
4200 PRINT "Y-MAX=";M2;"  Y-MIN=
";M1 @ PRINT "Y-RANGE=";M2-M1
4210 DISP "CHANGE Y-LIMITS?(Y/N)
" @ INPUT C3$
4220 IF C3$="N" THEN 4250
4230 DISP "NEW Y-MIN AND Y-MAX:"
 @ INPUT M1,M2
4240 PRINT "Y-PLOT LIMITS:";M1;M
2
4250 DISP "Y-AXIS MAJ&MIN TICS:"
 @ INPUT D4,D2
4260 S2=S1+D*3/2 @ G9=S2+(N-1)*D

4270 PRINT "X-MAX=";G9;"  X-MIN=
";S2 @ PRINT "X-RANGE=";G9-S2
4280 DISP "CHANGE X-LIMITS?(Y/N)
" @ INPUT C3$
4290 IF C3$="N" THEN 4320
4300 DISP "NEW X-MIN AND X-MAX:"
 @ INPUT S2,G9
4310 PRINT "X-PLOT LIMITS";S2;G9

4320 DISP "X-AXIS MAJ&MIN TICS:"
 @ INPUT D3,D1
4330 PRINT "XMJTCS=";D3;"  XMNTC
S=";D1 @ PRINT "YMJTCS=";D4;"  Y
MNTCS=";D2
4340 DISP "OVERLAP PLOTS?(Y/N)"
@ INPUT V2$
4350 IF V2$="N" THEN GCLEAR
4360 H9=(M2-M1)*.05 @ G9=G9-S2
4370 H1=M1+H9 @ G1=.05*G9+S2
4380 H2=M2-H9 @ G2=.88*G9+S2
4390 G9=S2+G9 @ SCALE S2,G9,M1,M
2
4400 PRINT "XXXXXXXXXXXXXXXXXXXX
XXX"
4410 PRINT "

    "
4420 DISP "TITLE OF Y-AXIS" @ IN
PUT L1$
4430 DISP "TITLE OF X-AXIS" @ IN
PUT L2$
4440 PRINT "            ";L1$
4450 XAXIS M1,D1,S2,G9
4460 YAXIS S2,D2,M1,M2
4470 ! LABEL X-AXIS
4480 LDIR 90
4490 FOR X=S2+D3 TO G9-D3 STEP D
3
4500 MOVE X,H1 @ LABEL VAL$(X)
4510 NEXT X
4520 ! LABEL Y-AXIS
4530 LDIR 0
4540 FOR Y=M1 TO M2 STEP D4
4550 MOVE G1,Y @ LABEL VAL$(Y)
4560 NEXT Y
4570 PENUP @ MOVE S2,W(1)
4580 S2=S2+(I5-1)*D/2
4590 FOR J=1 TO N
4600 X=S2+(J-1)*D
4610 DRAW X,W(J)
4620 NEXT J
4630 COPY
4640 PRINT L2$
4650 CLEAR
4660 RETURN
4670 ! SUBPROGRAM MAXMIN
4680 M1=W(1) @ M2=W(1)
4690 FOR I=2 TO N
4700 IF W(I)<M1 THEN M1=W(I)
4710 IF W(I)>M2 THEN M2=W(I)
4720 NEXT I
4730 RETURN
4740 ! SUBPROGRAM STOP
4750 PRINT "===================
"

4760 CLEAR @ DISP "ALL DONE!!!"
@ BEEP 100,200
4770 STOP
4780 ! SUBPROGRAM TRANSFER
4790 IF W2=1 THEN U1$="PSEUDO-RE
SISTIVITY"
4800 IF W2=2 THEN U1$="PSEUDO-CH
R.. OR PFE"
4810 IF W2=3 THEN U1$="PSEUDO-M.
FACTOR"
4820 IF W2=4 THEN U1$="PSEUDO-PH
ASE"
4830 FOR J=1 TO N
4840 FOR I=1 TO M
4850 IF W2=1 THEN Z(I,J)=R(I,J)
4860 IF W2=2 THEN Z(I,J)=F(I,J)
4870 IF W2=3 THEN Z(I,J)=M(I,J)
4880 IF W2=4 THEN Z(I,J)=P(I,J)
4890 NEXT I
4900 NEXT J
4910 RETURN
```

Appendix 2.--Examples of program operation

KEY (1) OPERATION MODE
EXAMPLE

```
THIS SUBPROGRAM ALLOWS DATA INPU
T FROM TAPE OR KEYBOARD
OPERATE KEY?(Y/N):
?
Y
INPUT FROM TAPE (T) OR KEYBOARD(
K):
?
K
AREA NAME?:
?
BAID AL JIMALAH
LINE NAME?
?
500E
TYPE OF SURVEY, TIME DOMAIN(T) O
R FREQ. DOMAIN(F)?
?
F
STATION LOCATION WILL BE -VE FOR
 S&W AND +VE FOR N&E
FIRST STATION LOCAION ON N=1 DIP
OLE?:
?
-100
STATION ORDER DECREASE (D) OR IN
CREASE (I)?:
?
I
NO. OF DIPOLES ALONG PROFILE?
?
4
DIPOLE LENGTH?:
?
50
NO OF IP LEVELS(N=)?:
?
4
 FREQUENCY DOMAIN IP SURVEY
AREA:BAID AL JIMALAH
LINE:500E  FIRST STATION:-25
NO. OF DIPOLES= 4     N= 4
DIPOLE LENGTH: 50
-----------------
INPUT DATA REDUCED?(Y/N):
?
N
ONE FREQUENCY USED?(Y/N)
?
Y
--INPUT DATA---
INPUT PSEUDO(P) OR WINDOW(W)?
?
P
DIPOLE NO.: 1
DIPOLE NO= 1  STATION=-25
N=    VP        CR     PFE    RPS
FOR LEVEL N= 1
INPUT VP,Cr,PFE,RPS
?
60,2.42,1.2,-.1
 1   60.0     2.4     1.2    -.1
FOR LEVEL N= 2
INPUT VP,Cr,PFE,RPS
?
32,2.36,2.3,-.8
 2   32.0     2.4     2.3    -.8
FOR LEVEL N= 3
INPUT VP,Cr,PFE,RPS
?
17,2.43,2.6,-.8
 3   17.0     2.4     2.6    -.8
FOR LEVEL N= 4
INPUT VP,Cr,PFE,RPS
?
10.9,2.57,1.2,-.5
 4   10.9     2.6     1.2    -.5
DIPOLE NO.: 2
DIPOLE NO= 2  STATION= 25
N=    VP        CR     PFE    RPS
FOR LEVEL N= 1
INPUT VP,Cr,PFE,RPS
?
141,2.37,1.9,-1.1
 1  141.0     2.4     1.9   -1.1
FOR LEVEL N= 2
INPUT VP,Cr,PFE,RPS
?
63,2.43,2.3,-1.2
 2   63.0     2.4     2.3   -1.2
FOR LEVEL N= 3
INPUT VP,Cr,PFE,RPS
?
37,2.56,1.6,-.6
 3   37.0     2.6     1.6    -.6
FOR LEVEL N= 4
INPUT VP,Cr,PFE,RPS
?
18,2.23,1.8,-.2
 4   18.0     2.2     1.8    -.2
DIPOLE NO.: 3
DIPOLE NO= 3  STATION= 75
N=    VP        CR     PFE    RPS
FOR LEVEL N= 1
INPUT VP,Cr,PFE,RPS
?
85,2.43,1.8,-0.8
 1   85.0     2.4     1.8    -.8
FOR LEVEL N= 2
INPUT VP,Cr,PFE,RPS
?
37,2.57,1.2,-.2
 2   37.0     2.6     1.2    -.2
FOR LEVEL N= 3
INPUT VP,Cr,PFE,RPS
?
15.7,2.24,1.1,-.2
 3   15.7     2.2     1.1    -.2
FOR LEVEL N= 4
INPUT VP,Cr,PFE,RPS
?
19.8,2.61,.8,-.3
 4   19.8     2.6      .8    -.3
```

Appendix 2.--Examples of program operations--Continued

```
DIPOLE NO.: 4
DIPOLE NO= 4  STATION= 125
N=     VP        CR      PFE    RPS
FOR LEVEL N= 1
INPUT Vp,Cr,PFE,RPS
?
105,2.57,.8,-.3
 1  105.0     2.6       .8    -.3
FOR LEVEL N= 2
INPUT Vp,Cr,PFE,RPS
?
32,2.24,.8,.2
 2   32.0     2.2       .8     .2
FOR LEVEL N= 3
INPUT Vp,Cr,PFE,RPS
?
17.2,2.62,.6,-.3
 3   17.2     2.6       .6    -.3
FOR LEVEL N= 4
INPUT Vp,Cr,PFE,RPS
?
12.2,2.64,.6,.2
 4   12.2     2.6       .6     .2
```

KEY (2) OPERATION MODE
EXAMPLE

```
THIS ROUTINE ALOWS STORE ANY PAR
T OR ALL OF INPUT AND/OR OUTPUT
DATA ON TAPE
OPERATE KEY?(Y/N):
?
Y
OUTPUT FILENAME AND NO. OF REC.?
?
IPDAT,20
STORE PSEUDO(P) OR WINDOW(W) DAT
A?:
?
P
--------------------
DATA STORED ON FILE:IPDAT
REDUCED PSEUDOSECTION OF
AREA:BAID AL JIMALAH
LINE:500E
```

KEY (3) OPERATION MODE
EXAMPLE

```
THIS ROUTINE PROVIDES PRINO
 OF PSEUDOSECTIONS OR FILTERED D
ATA
OPERATE KEY?(Y/N):
?
Y
PRINT PSEUDO(P) OR WINDOW(W)
?
P
IP FIELD DATA
DIPOLE: 1   STATION:-25
N=      VP      CR    PFE    RPS
  1    60.0    2.4    1.2    -.1
  2    32.0    2.4    2.3    -.8
  3    17.0    2.4    2.6    -.8
  4    10.9    2.6    1.2    -.5
DIPOLE: 2   STATION: 25
N=      VP      CR    PFE    RPS
  1   141.0    2.4    1.9   -1.1
  2    63.0    2.4    2.3   -1.2
  3    37.0    2.6    1.6    -.6
  4    18.0    2.2    1.8    -.2
DIPOLE: 3   STATION: 75
N=      VP      CR    PFE    RPS
  1    85.0    2.4    1.8    -.8
  2    37.0    2.6    1.2    -.2
  3    15.7    2.2    1.1    -.2
  4    19.8    2.6     .8    -.3
DIPOLE: 4   STATION: 125
N=      VP      CR    PFE    RPS
  1   105.0    2.6     .8    -.3
  2    32.0    2.2     .8     .2
  3    17.2    2.6     .6    -.3
  4    12.2    2.6     .6     .2
```

KEY (4) OPERATION MODE
EXAMPLE

```
THIS SUBPROGRAM ALLOWS CORRECTIO
N OF MISTAKES IN INPUT DATA
OPERATE KEY?(Y/N):
?
Y
ANY MISTAKES (Y/N)?
?
Y
MISTAKES IN PSEUDO?(Y/N)
?
Y
WHICH DIPOLE(J) AND N (I)
?
3,2
CORRECTION OF INPUT VALUE OF DIP
OLE: 3 N= 2
WRONG VALUE:
 37  2.57  1.2 -.2
ENTER NEW VALUES:
?
37,2.58,1.2,-.2
CORRECT VALUE:
 37  2.58  1.2 -.2
ANY MISTAKES (Y/N)?
?
N
```

KEY (5) OPERATION MODE
EXAMPLE

```
THIS ROUTINE ALLOWS REDUCION OF
PSEUDOSECTION DATA
OPERATE KEY?(Y/N)
?
Y
```

```
TO PRINTOUT REDUCED IP-DATA
USE KEY (3) OPERATION MODE
----------------------------


THIS ROUTINE PROVIDES PRINT OUTS
 OF PSEUDOSECTIONS OR FILTERED D
ATA
OPERATE KEY?(Y/N):
?
Y
PRINT PSEUDO(P) OR WINDOW(W)
?
P
PSEUDO SECTION DATA
DIPOLE: 1  STATION:-25
N=       RES.    PFE     MF      RPS
  1     23.4     1.2    51.4     -.1
  2     51.1     2.3    45.0     -.8
  3     65.9     2.6    39.4     -.8
  4     79.9     1.2    15.0     -.5
DIPOLE: 2  STATION: 25
N=       RES.    PFE     MF      RPS
  1     56.1     1:9    33.9    -1.1
  2     97.7     2.3    23.5    -1.2
  3    136.2     1.6    11.7     -.6
  4    152.1     1.8    11.8     -.2
DIPOLE: 3  STATION: 75
N=       RES.    PFE     MF      RPS
  1     33.0     1.8    54.6     -.8
  2     54.1     1.2    22.2     -.2
  3     66.1     1.1    16.7     -.2
  4    143.0      .8     5.6     -.3
DIPOLE: 4  STATION: 125
N=       RES     PFE     MF      RPS
  1     38.5      .8    20.8     -.3
  2     53.9      .8    14.9      .2
  3     61.9      .6     9.7     -.3
  4     87.1      .6     6.9      .2
```

KEY (6) OPERATION MODE
EXAMPLE

```
THIS ROUTINE ALLOWS APPLICATION
OF THREE IP FILTERS
SUGG BY FRASER (1981)
WINDOW (1) FOR AVER. LEVELS (2)
FOR TOTAL AVER AND (3) FOR AVERA
GE TRIANGLE SIDES
OPERATE KEY?(Y/N)
?
Y
APPLY MAP WINDOW?(Y/N)
?
Y
TITLE OF DATA TO BE FILTERED
?
RESISTIVITY
FILTERED MAP RESULTS OF
RESISTIVITY
AVERAGE LEVELS FILTER
----------------------
AV. LEV.(1) OR ALL AV.(2) OR TRI
ANGLE SIDES (3):
?
1
TO RES.(1),PFE(2),MF(3) OR RPS(4
)?
?
1
FILTERED MAP RESULTS OF
PSEUDO-RESISTIVITY
AVERAGE LEVELS FILTER
------------------
STATION                 MAP VALUE
  -25.0                     55.1
   25.0                     86.9
   75.0                     80.8
  125.0                     77.2
APPLY MAP WINDOW?(Y/N)
?
Y
TITLE OF DATA TO BE FILTERED
? -
PFE
AV. LEV.(1) OR ALL AV.(2) OR TRI
ANGLE SIDES (3):
?
2
TO RES.(1),PFE(2),MF(3) OR RPS(4
)?
?
2
FILTERED MAP RESULTS OF
PSEUDO-CHR. OR PFE
TOTAL AVERAGE FILTER
------------------
STATION                 MAP VALUE
  -25.0                      1.8
   25.0                      1.9
   75.0                      1.1
  125.0                      1.2
APPLY MAP WINDOW?(Y/N)
?
Y
TITLE OF DATA TO BE FILTERED
?
M.FACTOR
AV. LEV.(1) OR ALL AV.(2) OR TRI
ANGLE SIDES (3):
?
3
TO RES.(1),PFE(2),MF(3) OR RPS(4
)?
?
3
FILTERED MAP RESULTS OF
PSEUDO-M.FACTOR
AVERAGE SIDES FILTER
------------------
STATION                 MAP VALUE
  -25.0                     37.7
   25.0                     25.2
   75.0                     27.0
  125.0                     17.4
APPLY MAP WINDOW?(Y/N)
?
N
```

Appendix 2.--Examples of program operations--Continued

KEY (7) OPERATION MODE
EXAMPLE

```
THIS ROUTINE ALOWS X-Y PLOTS OF
IP OUTPUT DATA ALONG PROFILE IN
LOG OR NORM  SCALE
OPERATE KEY?(Y/N)
?
Y
NORMAL(N) OR LOG(L) PLOT?:
?
N
PLOT WINDOW(W) OR PSEUDO LEVEL(P
)
?
W
-------------------
PLOT OF LINE:500E
AREA:BAID AL JIMALAH
PLOT PARAMETERS:
Y-MAX= 38   Y-MIN= 17
Y-RANGE= 21
CHANGE Y-LIMITS?(Y/N)
?
Y
NEW Y-MIN AND Y-MAX:
?
10,50
Y-PLOT LIMITS: 10  50
Y-AXIS MAJ&MIN TICS:
?
10,2
X-MAX= 125   X-MIN=-25
X-RANGE= 150
CHANGE X-LIMITS?(Y/N)
?
Y
NEW X-MIN AND X-MAX:
?
-50,150
X-PLOT LIMITS-50  150
X-AXIS MAJ&MIN TICS:
?
50,10
XMJTCS= 50   XMNTCS= 10
YMJTCS= 10   YMNTCS= 2
OVERLAP PLOTS?(Y/N)
?
N
XXXXXXXXXXXXXXXXXXXXXX


TITLE OF Y-AXIS
?
M.FACTOR
TITLE OF X-AXIS
?
METERS
```